This Maple worksheet contains the computations for the paper "*Unidirectional migration of populations with Allee effect*";
- the simple example in Section 3,
- the main computations of Sections 4 and 5 where the parameter region is partitioned according to the number of non-negative steady states that the system in equation (1) of the paper has.

Example of Section 3

First defining the polynomial f of Section 3. Note that in algebra we do not need to define a polynomial as a function in analysis, we just assign a name to the polynomial expression and that is enough.

```
> f := x*(1-x)*(x-b);
```

$$f := x\,(1-x)\,(x-b) \tag{1.1}$$

To take derivative of "f" with respect to "x" in Maple, simply use "diff(f, x)". The result is again a polynomial. Now we want to use the Maple command "EliminationIdeal" of the Maple package "PolynomialIdelas" to get the projection of the solution set of "f = diff(f, x) = 0" when it is seen as a system in both x and b, into the b-axis.

```
> J := PolynomialIdeals:-EliminationIdeal( PolynomialIdeals:-
  PolynomialIdeal( f, diff( f, x ) ), { b } );
```

$$J := \langle b^4 - 2\,b^3 + b^2 \rangle \tag{1.2}$$

The output is in the form of an ideal, we just need the generating polynomials of this ideal which we can see in the above output. For the interested reader, one can use the command "Generators" from the same package to extract the generating polynomials in a set data structure.

```
> PolynomialIdeals:-Generators( J );
```

$$\{b^4 - 2\,b^3 + b^2\} \tag{1.3}$$

It is a single polynomial. To factorize a polynomial, one can use the Maple command "factor" which is accessible without the need of calling a specific package.

```
> factor( b^4-2*b^3+b^2 );
```

$$b^2\,(b-1)^2 \tag{1.4}$$

Both way migration.

Consider the parametric system of polynomial equations given at equation (2) of the paper with the inequality constraints on the variables and the parameters, and \delta_{1,2}=\delta_{2,1}=1.

We define F to be the list of the two polynomial equations resulted by letting \dot{N_1}=\dot_{N_2}=0.

```
> F := [ N[1]*(1-N[1])*(N[1]-b)-a*N[1]+a*N[2], N[2]*(1-N[2])*(N[2]-b)+a*N[1]-a*N[2] ]:
  < seq( i, i in F ) >;
```

$$\begin{bmatrix} N_1 (1 - N_1) (N_1 - b) - a N_1 + a N_2 \\ N_2 (1 - N_2) (N_2 - b) + a N_1 - a N_2 \end{bmatrix} \quad \textbf{(2.1)}$$

One can easily check that the number of nonnegative solutions for F=0 when "b=a=0" or "b=1 and a=0" is 4, and when "b=0 and a \neq 0" or "b=1 and a \neq 0" is 2.

If a,b > 0, then N_1=0 together with F=0 implies N_2=0 and vice versa, N_2=0 together with F=0 implies N_1=0. Therefore for every a,b>0, (0,0) is one nonnegative solution to the system, and the only boundary solution. Therefore we can reduce the constraints to N_1,N_2,a,b > 0 which are acceptable constraints for the Maple package **RootFinding:-Parametric**. Note that this package uses an open CAD algorithm, therefore non-strict inequalities are not acceptable inputs for its commands.

The following "cadTwoSidedModel" stores the result of "CAD with respect to the discriminant variety" of the parametric system made by F=0 and the positivity constraints on the variables and the parameters.

```
> cadTwoSidedModel := RootFinding:-Parametric:-CellDecomposition( [ seq( i = 0, i in F ), seq( N[i] > 0, i = 1 .. 2 ), a > 0, b > 0 ], [ seq( N[i], i = 1 .. 2 ) ], [ a, b ] ):
```

Up to here Maple has decomposed the positive orthant of the parameter space to a finite number of disjoint open connected sets so that the number of positive solutions to the system F=0 is invariant on each of these sets. Note that there might be more than one open region in this decomposition with the same number of solutions.

ListNumberSolutions is a list which its i-th member is the number of positive solutions to the system F=0 on the i-th open region in the above decomposition. Then we ask Maple to print the converted version of this list to a set. Recall that a Maple set data structure does not allow repitition while a Maple list data structure does.

```
> ListNumberSolutions := RootFinding:-Parametric:-NumberOfSolutions( cadTwoSidedModel ):
  { seq( i[2], i in ListNumberSolutions ) };
```

$$\{2, 4, 8\} \quad \textbf{(2.2)}$$

Therefore the number of possible positive solutions to the system on open regions in the positive orthant of the parameter space are 2, 4 and 8. Together with the only boundary soultion, i.e. (0,0), the sytem can have 3, 5 or 9 nonnegative solutions.

cellsIndexTable is a table (Maple's table data structure is similar to Python's dictionary data structure) storing index of the open regions in the decomposition with specific number of solutions.

```
> cellsIndexTable := table( [ seq( i = RootFinding:-Parametric:-CellsWithSolutions( cadTwoSidedModel, i ), i in (2.2) ) ] );
```

$$cellsIndexTable := \mathrm{table}([2 = [3, 6, 9, 12], 4 = [2, 5, 8, 11], 8 = [1, 4, 7, 10]]) \tag{2.3}$$

In below we ask Maple to plot these regions and color them with red, orange and yellow if the number of positive solutions on them are 8, 4 or 2 respectively.

```
> A2 := Array( 1 .. numelems( cellsIndexTable[2] ), 0 ):
  A4 := Array( 1 .. numelems( cellsIndexTable[4] ), 0 ):
  A8 := Array( 1 .. numelems( cellsIndexTable[8] ), 0 ):
  for i to numelems( cellsIndexTable[2] ) do
      A2[i] := RootFinding:-Parametric:-CellPlot(
  cadTwoSidedModel, cellsIndexTable[2][i], color = yellow ):
  end do:
  for i to numelems( cellsIndexTable[4] ) do
      A4[i] := RootFinding:-Parametric:-CellPlot(
  cadTwoSidedModel, cellsIndexTable[4][i], color = coral ):
  end do:
  for i to numelems( cellsIndexTable[8] ) do
      A8[i] := RootFinding:-Parametric:-CellPlot(
  cadTwoSidedModel, cellsIndexTable[8][i], color = red ):
  end do:
  PTwoSidedModel := plots:-display( seq( A2[i], i = 1 .. numelems
  ( cellsIndexTable[2] ) ), seq( A4[i], i = 1 .. numelems(
  cellsIndexTable[4] ) ), seq( A8[i], i = 1 .. numelems(
  cellsIndexTable[8] ) ) );
```



As one can see from this figure, the decomposition of the parameter space with respect to the number of non-negative solutions to the system F=0, has symmetries with respect to b=1/2 and b=1. Which we already were aware of it using Lemmas 2.1 and 2.2 from the reference [Röst, G. and A. Sadeghimanesh (2021)] of the paper.

```
> plots:-display( PTwoSidedModel, view = [ 0 .. 0.15, 0 .. 0.5 ],
  'labels' = [ a, b ], 'labelfont' = [ "TimesNewRoman", 20 ] );
```




```
> text1 := plots:-textplot( [ 0.01, 0.4, "9" ], 'align' = {
  'above', 'right' }, 'color' = 'black', 'font' = [
  "TimesNewRoman", 30 ] ):
  text2 := plots:-textplot( [ 0.05, 0.25, "5" ], 'align' = {
  'above', 'right' }, 'color' = 'black', 'font' = [
  "TimesNewRoman", 30 ] ):
  text3 := plots:-textplot( [ 0.12, 0.15, "3" ], 'align' = {
  'above', 'right' }, 'color' = 'black', 'font' = [
  "TimesNewRoman", 30 ] ):
  plots:-display( PTwoSidedModel, text1, text2, text3, view = [ 0
  .. 0.15, 0 .. 0.5 ], 'labels' = [ a, b ], 'labelfont' = [
  "TimesNewRoman", 20 ] );
```



This is Figure 2 of the paper.

The boundary between these regions as explained in Section 3 of the paper is defined by the discriminant variety. Note that Maple already has computed the discriminant variety when we defined "cadTwoSidedModel" using "RootFinding:-Parametric:-CellDecomposition" command. To access the information on this discriminant variety use ":-DiscriminantVariety" after "cadTwoSidedModel" as in below.

```
> cadTwoSidedModel:-DiscriminantVariety;
```

$$[[a], [b], [2a+b], [b-1], [2a-b+1], [b^2+2a-b], [4ab^4-36a^2b^2-8ab^3-b^4+108a^3+36a^2b+12ab^2+2b^3-36a^2-8ab-b^2+4a]] \quad (2.4)$$

The equations a = 0, b = 0, b - 1 = 0 are just the equations defining the parameter region we had given to Maple ourselves, i.e. a > 0 and 0 < b < 1.
Also note that the equation 2*a + b = 0 has no solution in the positive orthant so we can simply ignore it.
There are three equations left.

```
> equation1 := 2*a - b + 1;
  equation2 := b^2 + 2*a - b;
  equation3 := 4*a*b^4 - 36*a^2*b^2 - 8*a*b^3 - b^4 + 108*a^3 +
  36*a^2*b + 12*a*b^2 + 2*b^3 - 36*a^2 - 8*a*b - b^2 + 4*a;
```

$$equation1 := 2\,a - b + 1$$

$$equation2 := b^2 + 2\,a - b$$

$$equation3 := 4\,a\,b^4 - 36\,a^2\,b^2 - 8\,a\,b^3 - b^4 + 108\,a^3 + 36\,b\,a^2 + 12\,a\,b^2 + 2\,b^3 - 36\,a^2 - 8\,b\,a - b^2 + 4\,a \quad \textbf{(2.5)}$$

Equation 3 defines the boundary between the regions with 9 and 5 steady states, depicted below in red color. Equation 2, colored in blue, defines the boundary between the regions with 5 and 3 steady states when 0 < b < 1, this boundary when b > 1 is defined by equation 1, colored in green, which is a transformation of equation 1 under the symmetry, (a, b) --> (a / b^2, 1 / b), found in Lemma 2.1 of Röst and Sadeghimanesh (2021). Also note that Equation 2 has the symmetry found in Lemma 2.2 of Röst and Sadeghimanesh (2021), (a, b) --> (a, 1 - b). And equation 3 has both symmetries.

```
> PEq1 := plots:-implicitplot( equation1 = 0, a = 0 .. 0.4, b = 0
  .. 3, view = [ 0 .. 0.4, 0 .. 3 ], color = green ):
  PEq2 := plots:-implicitplot( equation2 = 0, a = 0 .. 0.4, b = 0
  .. 3, view = [ 0 .. 0.4, 0 .. 3 ], color = blue ):
  PEq3 := plots:-implicitplot( equation3 = 0, a = 0 .. 0.4, b = 0
  .. 3, view = [ 0 .. 0.4, 0 .. 3 ], color = red ):
  plots:-display( PEq1, PEq2, PEq3, view = [ 0 .. 0.4, 0 .. 3 ],
  'labels' = [ a, b ], 'labelfont' = [ "TimesNewRoman", 20 ] );
```



One way migration.

Consider the parametric system of polynomial equations given at equation (2) of the paper with the inequality constraints on the variables and the parameters ,and \delta_{1,2}=1, \delta_{2,1}=0.

We define F to be the list of the two polynomial equations resulted by letting \dot{N_1}=\dot_{N_2}=0 of equation (1).

```
> F := [ N[1]*(1-N[1])*(N[1]-b)-a*N[1], N[2]*(1-N[2])*(N[2]-b)+a*
  N[1] ]:
  < seq( i, i in F ) >;
```

$$\begin{bmatrix} N_1 (1 - N_1) (N_1 - b) - a N_1 \\ N_2 (1 - N_2) (N_2 - b) + a N_1 \end{bmatrix} \tag{3.1}$$

One can easily check that the number of nonnegative solutions for F=0 when "b=a=0" or "b=1 and a=0" is 4, and when "b=0 and a \neq 0" or "b=1 and a \neq 0" is 2.

If a,b > 0, then N_1=0 together with F=0 implies N_2 = 0 or 1 or b. If a,b > 0, then N_2=0 together with F=0 implies N_1=0. Therefore for every a,b>0, (0,0), (0,b) and (0,1) are nonnegative solutions to the system, and the only boundary solutions. Therefore we can reduce the constraints to N_1,N_2, a,b > 0 which are acceptable constraints for the maple package RootFinding[Parametric]. Note that this package uses an open CAD algorithm, therefore non-strict inequalities are not acceptable inputs for its predefined functions.

The following "cadOneSidedModel" stores the result of "CAD with respect to the discriminant variety" of the parametric system made by F=0 and the positivity constraints on the variables and the parameters.

```
> cadOneSidedModel := RootFinding:-Parametric:-CellDecomposition(
  [ seq( i = 0, i in F ), seq( N[i] > 0, i = 1 .. 2 ), a > 0, b >
  0 ], [ seq( N[i], i = 1 .. 2 ) ], [ a, b ] ):
```

Up to here Maple has decomposed the positive orthant of the parameter space to a finite number of disjoint open connected sets so that the number of positive solutions to the system F=0 is invariant on each of these sets. Note that there might be more than one open region in this decomposition with the same number of solutions.

ListNumberSolutions is a list which its i-th member is the number of positive solutions to the system F=0 on the i-th open region in the above decomposition. Then we ask Maple to print the converted version of this list to a set.

```
> ListNumberSolutions := RootFinding:-Parametric:-
  NumberOfSolutions( cadOneSidedModel ):
  { seq( i[2], i in ListNumberSolutions ) };
```

$$\{0, 2, 4, 6\} \qquad \mathbf{(3.2)}$$

Therefore the number of possible positive solutions to the system on open regions in the positive orthant of the parameter space are 0, 2, 4 and 6. Together with the three boundary soultion, the sytem can have 3, 5, 7 or 9 nonnegative solutions.

cellsIndexTable is a table (Maple's table data structure is similar to Python's dictionary data structure) storing index of the open regions in the decomposition with specific number of solutions.

```
> cellsIndexTable := table( [ seq( i = RootFinding:-Parametric:-
  CellsWithSolutions( cadOneSidedModel, i ), i in (3.2) ) ] );
```

$$\mathit{cellsIndexTable} := \mathrm{table}([0 = [4, 8, 12, 16, 18, 20, 24, 28, 32, 36], 2 = [3, 7, 11, 27, 31, 35], 4 = [2, 6, 10, 14, 22, 26, 30, 34], 6 = [1, 5, 9, 13, 15, 17, 19, 21, 23, 25, 29, 33]]) \qquad \mathbf{(3.3)}$$

In below we ask Maple to plot these regions and color them with red, brown, orange and yellow if the number of positive solutions on them are 6, 4, 2 or 0 respectively.

```
> A0 := Array( 1 .. numelems( cellsIndexTable[0] ), 0 ):
  A2 := Array( 1 .. numelems( cellsIndexTable[2] ), 0 ):
  A4 := Array( 1 .. numelems( cellsIndexTable[4] ), 0 ):
  A6 := Array( 1 .. numelems( cellsIndexTable[6] ), 0 ):
```

```
for i to numelems( cellsIndexTable[0] ) do
    A0[i] := RootFinding:-Parametric:-CellPlot(
cadOneSidedModel, cellsIndexTable[0][i], color = yellow ):
end do:
for i to numelems( cellsIndexTable[2] ) do
    A2[i] := RootFinding:-Parametric:-CellPlot(
cadOneSidedModel, cellsIndexTable[2][i], color = coral ):
end do:
for i to numelems( cellsIndexTable[4] ) do
    A4[i] := RootFinding:-Parametric:-CellPlot(
cadOneSidedModel, cellsIndexTable[4][i], color = orange ):
end do:
for i to numelems( cellsIndexTable[6] ) do
    A6[i] := RootFinding:-Parametric:-CellPlot(
cadOneSidedModel, cellsIndexTable[6][i], color = red ):
end do:
POneSidedModel := plots:-display( seq( A0[i], i = 1 .. numelems
( cellsIndexTable[0] ) ), seq( A2[i], i = 1 .. numelems(
cellsIndexTable[2] ) ),seq( A4[i], i = 1 .. numelems(
cellsIndexTable[4] ) ), seq( A6[i], i = 1 .. numelems(
cellsIndexTable[6] ) ) );
```



As one can see from this figure, the decomposition of the parameter space with respect to the number of non-negative solutions to the system F=0, has only one symmetry with respect to and b=1 and there is no symmetry with respect to b=1/2. Which we already were aware of it. The symmetry with resect to b=1 is a result of Lemma 4.1 of the paper.

```
> cadPlotFull := plots:-display( POneSidedModel, view = [ 0 ..
  0.3, 0 .. 1 ], 'labels' = [ a, b ], 'labelfont' = [
  "TimesNewRoman", 20 ] ):
  cadPlotZoomed := plots:-display( POneSidedModel, view = [ 0.057
  .. 0.069, 0.49 .. 0.505 ], 'labels' = [ a, b ], 'labelfont' = [
  "TimesNewRoman", 20 ] ):
  resulto := DocumentTools:-Tabulate( [ cadPlotFull,
  cadPlotZoomed ], exterior = none, interior = none, weights = [
  100, 100 ], widthmode = pixels, width = 600);
```

$$resulto := \text{"Tabulate"} \qquad \textbf{(3.4)}$$




```
> textL1 := plots:-textplot( [ 0.01, 0.5, "9" ], 'align' = {
  'above', 'right' }, 'color' = 'black', 'font' = [
  "TimesNewRoman", 30 ] ):
  textL2 := plots:-textplot( [ 0.025, 0.24, "7" ], 'align' = {
  'above', 'right' }, 'color' = 'black', 'font' = [
  "TimesNewRoman", 30 ] ):
  textL3 := plots:-textplot( [ 0.08, 0.2, "5" ], 'align' = {
  'above', 'right' }, 'color' = 'black', 'font' = [
  "TimesNewRoman", 30 ] ):
  textL4 := plots:-textplot( [ 0.22, 0.3, "3" ], 'align' = {
  'above', 'right' }, 'color' = 'black', 'font' = [
  "TimesNewRoman", 30 ] ):

  textR1 := plots:-textplot( [ 0.0585, 0.502, "9" ], 'align' = {
```

```
'above', 'right' }, 'color' = 'black', 'font' = [
"TimesNewRoman", 30 ] ):
textR2 := plots:-textplot( [ 0.059, 0.493, "7" ], 'align' = {
'above', 'right' }, 'color' = 'black', 'font' = [
"TimesNewRoman", 30 ] ):
textR3 := plots:-textplot( [ 0.064, 0.490, "5" ], 'align' = {
'above', 'right' }, 'color' = 'black', 'font' = [
"TimesNewRoman", 30 ] ):
textR4 := plots:-textplot( [ 0.066, 0.494, "3" ], 'align' = {
'above', 'right' }, 'color' = 'black', 'font' = [
"TimesNewRoman", 30 ] ):

cadPlotFull := plots:-display( POneSidedModel, textL1, textL2,
textL3, textL4, view = [ 0 .. 0.3, 0 .. 1 ], 'labels' = [ a, b
], 'labelfont' = [ "TimesNewRoman", 20 ] ):
cadPlotZoomed := plots:-display( POneSidedModel, textR1,
textR2, textR3, textR4, view = [ 0.057 .. 0.069, 0.49 ..
0.505], 'labels' = [ a, b ],'labelfont' = [ "TimesNewRoman", 20
], tickmarks = [ default, [ 0.490 = "0.490 ", 0.495 = "0.495 ",
0.4961661 = cat( 'beta__1', "   " ), 0.500 = "0.500 ", 0.505 =
"0.505 " ] ] ):
DocumentTools:-Tabulate( [ cadPlotFull, cadPlotZoomed ],
exterior = none, interior = none, weights = [ 100, 100 ],
widthmode = pixels, width = 600 );
```

"Tabulate0" **(3.5)**



This is Figure 3 of the paper.

Remember that the boundary between these regions as explained in Section 3 of the paper is defined by the discriminant variety and that Maple already has computed the discriminant variety when we defined "cadOneSidedModel" using "RootFinding:-Parametric:-CellDecomposition" command.

```
> cadOneSidedModel:-DiscriminantVariety;
```

$$[[a], [b], [a+b], [-b^2+4a+2b-1], [16a^3b^6+16a^2b^7-4ab^8+108a^4b^4+60a^3b^5-75a^2b^6+10ab^7+b^8-54a^4b^3-12a^3b^4+42a^2b^5+4ab^6-4b^7+729a^6+1458a^5b+405a^4b^2-328a^3b^3+50a^2b^4-20ab^5+6b^6-54a^4b-12a^3b^2+42a^2b^3+4ab^4-4b^5+108a^4+60a^3b-75a^2b^2+10ab^3+b^4+16a^3+16a^2b-4ab^2]] \quad (3.6)$$

The equations a = 0 and b = 0 are two of the constraints we added ourselves from the beginning.
The equation a + b = 0 has no solution in the positive orthant and can be ignored.
There are only two equations left which are important.

```
> equation1 := 16*a^3*b^6 + 16*a^2*b^7 - 4*a*b^8 + 108*a^4*b^4 +
  60*a^3*b^5 - 75*a^2*b^6 + 10*a*b^7 + b^8 - 54*a^4*b^3 - 12*a^3*
  b^4 + 42*a^2*b^5 + 4*a*b^6 - 4*b^7 + 729*a^6 + 1458*a^5*b +
  405*a^4*b^2 - 328*a^3*b^3 + 50*a^2*b^4 - 20*a*b^5 + 6*b^6 - 54*
  a^4*b - 12*a^3*b^2 + 42*a^2*b^3 + 4*a*b^4 - 4*b^5 + 108*a^4 +
  60*a^3*b - 75*a^2*b^2 + 10*a*b^3 + b^4 + 16*a^3 + 16*a^2*b - 4*
  a*b^2;
  equation2 := -b^2 + 4*a + 2*b - 1;
```

$$equation1 := 16a^3b^6+16a^2b^7-4ab^8+108a^4b^4+60a^3b^5-75a^2b^6+10ab^7+b^8-54a^4b^3-12a^3b^4+42a^2b^5+4ab^6-4b^7+729a^6+1458a^5b+405a^4b^2-328a^3b^3+50a^2b^4-20ab^5+6b^6-54a^4b-12a^3b^2+42a^2b^3+4ab^4-4b^5+108a^4+60a^3b-75a^2b^2+10ab^3+b^4+16a^3+16a^2b-4ab^2$$

$$equation2 := -b^2+4a+2b-1 \quad (3.7)$$

We will use these two equations in the Mathematica files as well. "equation1" and "equation2" in the files "Mathematica_two_connected_populations_with_Allee_effect_1" and "Mathematica_two_connected_populations_with_Allee_effect_2" are exactly the same expressions in above.

Equation 1, colored in red below, is defining the boundary containing the region with 7 steady states, and equation 2, colored in blue, is defining the boundary seperating the region with 3 steady states. One can easily see the symmetry found in Lemma 2.1 of Röst and Sadeghimanesh (2021) in both equations.

```
> PEq1 := plots:-implicitplot( equation1 = 0, a = 0 .. 0.4, b = 0
  .. 3, view = [ 0 .. 0.4, 0 .. 3 ], color = red ):
  PEq2 := plots:-implicitplot( equation2 = 0, a = 0 .. 0.4, b = 0
```

```
.. 3, view = [ 0 .. 0.4, 0 .. 3 ], color = blue ):
plots:-display( PEq1, PEq2, view = [ 0 .. 0.4, 0 .. 3 ],
'labels' = [ a, b ], 'labelfont' = [ "TimesNewRoman", 20 ] );
```



The two beta values reported in paper, beta1 and beta2, are where the bottom red curve has a horizontal tangent line, and the intersection of the red and blue curve when $0 < b < 1$, respectively. One can easily compute these two values. However, the information regarding these points are also computed in the CAD process.

The b-values of the points where the curves in the discriminant variety have tangent line parallel to the a-axis, or the intersection points of these curves, or their singular points are all among the roots of the univariate polynomials in b in the list of projection polynomials of the CAD object. To get these polynomials one can call ":-ProjectionPolynomials[1]" after "cadOneSidedModel" as in below.

```
> cadOneSidedModel:-ProjectionPolynomials[1];
```

$$\left[b, b-2, b-1, b+1, 2b-1, 59b^4-16b^3-214b^2-16b+59, 64b^6-912b^5+2823b^4-4066b^3+2823b^2-912b+64, b^4-7b^3-17b^2-7b+1\right] \quad \textbf{(3.8)}$$

The reader should be careful that we said the critical values of "b" are AMONG the roots of these projection polynomials, we did not say that all roots of these polynomials are critical values of b.

These polynomials restrict the search domain for finding the critical values to a finite set consisted of these roots, but one still needs to do manual checks to discard the extra values in this set.

b = 0 and b - 1 = 0 give us b = 0 and b = 1.
b - 2 = 0 is not in the region 0 < b < 1 (it is coming from Lemma 2.1 of Röst and Sadeghimanesh (2021) and having b = 1/2 as another critical value of b which we will see it in a second).
b + 1 = 0 has no non-negative solution so we ignore it.
2*b - 1 = 0 gives us b = 1/2 which is indeed a critical value as mentioned in the text and can be seen in the figure.
$59\,b^4 - 16\,b^3 - 214\,b^2 - 16\,b + 59 = 0$ is the polynomial that beta1 is a root of it. Note that it has only one real root satisfying 0 < b < 1. This can be checked by "sturm" command as in below which uses the Sturm sequence theorem to count the number of real roots of a univariate polynomial in an interval.

```
> sturm( 59*b^4 - 16*b^3 - 214*b^2 - 16*b + 59, b, 0, 1 );
```

$$1 \tag{3.9}$$

In fact using the same command (see below) one can see that this polynomial has two positive roots, and thus we have the sentence "beta1 is an algebraic number, specifically the smallest positive real root of 59*b^4 - 16*b^3 - 214*b^2 - 16*b + 59" in Section 5.

```
> sturm( 59*b^4 - 16*b^3 - 214*b^2 - 16*b + 59, b, 0, infinity );
```

$$2 \tag{3.10}$$

To get the approximate value of this number with 7 digits after the decimal point see the following line. The first line refines the interval (0, 1) to a smaller interval containing the root so that the two ends of the interval are equal until 7 digits after the decimal point. The second line shows the decimal representation of the two rational numbers of the ends of the new interval.

```
> RootFinding:-RefineRoot( 0 .. 1, 59*b^4 - 16*b^3 - 214*b^2 - 
  16*b + 59, b, digits = 7 );
```

$$\frac{36610596449513509417}{73786976294838206464} .. \frac{73221192899027018835}{147573952589676412928} \tag{3.11}$$

```
> evalf( (3.11) );
```

$$0.4961661026 .. 0.4961661026 \tag{3.12}$$

And that is why in the text it is said that beta1 with seven digits accuracy is 0.4961661.

The next equation is

64*b^6 - 912*b^5 + 2823*b^4 - 4066*b^3 + 2823*b^2 - 912*b + 64 = 0, this equation has one real root in the interval 0 < b < 1.

```
> sturm( 64*b^6 - 912*b^5 + 2823*b^4 - 4066*b^3 + 2823*b^2 - 912*
  b + 64, b, 0, 1 );
```

(3.13)

$$1 \tag{3.13}$$

```
> val := RootFinding:-RefineRoot( 0 .. 1, 64*b^6 - 912*b^5 +
  2823*b^4 - 4066*b^3 + 2823*b^2 - 912*b + 64, b, digits = 7 ):
  evalf( op[1]( val )[1] ); # op[1]( val )[1] takes the beginning
  of the interval found by RootFinding:-RefineRoot
```

$$0.0941096838 \tag{3.14}$$

One can easily see that this value is not a critical value for b. To test that no intersection point of equation1 and equation2 has such a b-coordinate, it is enough to substitute this value in both equation and then solve for a. Or equivalently solving the following system.

```
> # method 1
  solve( eval( [ equation1 = 0, equation2 = 0 ], b = RootOf( 64*
  b^6 - 912*b^5 + 2823*b^4 - 4066*b^3 + 2823*b^2 - 912*b + 64, b,
  0 .. 1 ) ), a, real );
> # method 2
  solve( [ equation1 = 0, equation2 = 0, 64*b^6 - 912*b^5 + 2823*
  b^4 - 4066*b^3 + 2823*b^2 - 912*b + 64 = 0, 0 < b, b < 1, a > 0
  ], [ a, b ] );
```

$$[\,] \tag{3.15}$$

whereas for beta1 value the solution set was not empty.

```
> solve( [ equation1 = 0, equation2 = 0, 59*b^4 - 16*b^3 - 214*
  b^2 - 16*b + 59 = 0, 0 < b, b < 1, a > 0 ], [ a, b ] );
```

$$\Big[\Big[a = RootOf\big(3481\,\_Z^4 - 9386\,\_Z^3 + 6625\,\_Z^2 - 1392\,\_Z + 64, 0.0625000000\,..0.0634800000\big), b = \frac{1}{11144}\big(191455\,RootOf\big(3481\,\_Z^4 - 9386\,\_Z^3 + 6625\,\_Z^2 - 1392\,\_Z + 64, 0.0625000000\,..0.0634800000\big)^3\big) - \frac{1}{5572}\big(230267\,RootOf\big(3481\,\_Z^4 - 9386\,\_Z^3 + 6625\,\_Z^2 - 1392\,\_Z + 64, 0.0625000000\,..0.0634800000\big)^2\big) + \frac{1}{11144}\big(218447\,RootOf\big(3481\,\_Z^4 - 9386\,\_Z^3 + 6625\,\_Z^2 - 1392\,\_Z + 64, 0.0625000000\,..0.0634800000\big)\big) - \frac{816}{1393}\Big]\Big] \tag{3.16}$$

The reader should not get confused by the above solution, the b value is the same value of beta1, just written with a different equivalent expression. As a quick check, one can try "evalf" command.

```
> evalf( (3.16) );
```

$$[[a = 0.0634621490, b = 0.4961661028]] \tag{3.17}$$

Going back to the value that we want to reject it being a critical value, other than showing it is not an intersection point, one should show that it is not a point with a horizontal tangent line or being a singular point of the two equations.

To show it is not a b-value of a point on equation1 with horizontal tangent line we should show that no point with such a b-coordinate can satisfy equation1 and its partial derivative with respect to a (but not vanishing the partial derivative with respect to b).

```
> solve( [ equation1 = 0, diff( equation1, a ) = 0, 64*b^6 - 912*
  b^5 + 2823*b^4 - 4066*b^3 + 2823*b^2 - 912*b + 64 = 0, 0 < b, b
  < 1, a > 0 ], [ a, b ] );
```

$$[\,] \tag{3.18}$$

whereas for b = 1/2 the solution set was not empty.

```
> solve( [ equation1 = 0, diff( equation1, a ) = 0, 2*b - 1 = 0,
  0 < b, b < 1, a > 0 ], [ a, b ] );
```

$$\left[\left[a = -\frac{1}{12} + \frac{\sqrt{3}}{12}, b = \frac{1}{2}\right]\right] \tag{3.19}$$

To show that it is not a b-value of a singular point on equation1 we should show that no point with such a b-coordinate can satisfy equation1 and both of its partial derivatives, but this is now trivial because such a point must satisfy the previous system of equations as well.

Now the last equation to check.

$$b^4 - 7\,b^3 - 17\,b^2 - 7\,b + 1 = 0$$

```
> sturm( b^4 - 7*b^3 - 17*b^2 - 7*b + 1, b, 0, 1 );
```

$$1 \tag{3.20}$$

```
> val := RootFinding:-RefineRoot( 0 .. 1, b^4 - 7*b^3 - 17*b^2 -
  7*b + 1, b, digits = 7 ):
  evalf( op[1]( val )[1] ); # op[1]( val )[1] takes the beginning
  of the interval found by RootFinding:-RefineRoot
```

$$0.1113735016 \tag{3.21}$$

One can easily see that this value is not a critical value for b. To test that no intersection point of equation1 and equation2 has such a b-coordinate, it is enough to substitute this value in both equation and then solve for a. Or equivalently solving the following system.

```
> # method 1
  solve( eval( [ equation1 = 0, equation2 = 0 ], b = RootOf( b^4
  - 7*b^3 - 17*b^2 - 7*b + 1, b, 0 .. 1 ) ), a, real );
> # method 2
  solve( [ equation1 = 0, equation2 = 0, b^4 - 7*b^3 - 17*b^2 -
  7*b + 1 = 0, 0 < b, b < 1, a > 0 ], [ a, b ] );
```

[ ] **(3.22)**

Now checking the horizontal tangent line case.

```
> solve( [ equation1 = 0, diff( equation1, a ) = 0, b^4 - 7*b^3 -
  17*b^2 - 7*b + 1 = 0, 0 < b, b < 1, a > 0 ], [ a, b ] );
```

[ ] **(3.23)**

So the final conclusion is that there are 4 critical values of b between 0 and 1: 0 and 1 themselves, 1/2 and beta1.

End of the file.

Source paper: Gergely Röst and AmirHosein Sadeghimanesh, Unidirectional migration of populations with Allee effect.